
Première Partie : 
Le Rendeur de monnaie 

Sommaire : 
 
 - Présentation de l’ensemble du système et diagramme sagital, 

 - Notice d’utilisation. 

Présentation de l’ensemble 

du système : 

 Le rendeur de monnaie est un sous système destiné à être associé à un 

système hôte afin de simplifier la restitution de la monnaie à l’usager. 

 

 Nous étudierons le rendeur de monnaie utilisé dans les automates 

d’affranchissement en libre service de la poste : 

 

 Dans ce système il existe trois rendeurs de monnaie qui permettent la 

restitution de pièces de 2 Francs, 50 centimes et 10 centimes. Il est alimenté par 

le secteur.Le rendeur de monnaie génère des tensions de 8 volts, 12 volts, 24 

volts et deux autres de 5 volts pour son fonctionnement interne.Un moto-

réducteur entraîne une noria qui élève les pièces jusqu’à la fenêtre de sortie où 

leur passage est détecté par deux capteurs optiques A et B. Le rendeur de 

monnaie est compatible avec differentes technologies de circuits logiques et 

gère differents modes de fonctionnement fixés par les signaux IN1 et IN2. Il ne 

nécessite pas d’entretient, ces indicateurs visuels facilitent la recherche de 

disfonctionnement. 

 

Caractèristiques du rendeur de monnaie : 
 

 - Le paiement des pièces de 20 à 30mm de diamètre s’effectue en 3 pièces 

par secondes, 

 - Détecte la fin de la réserve de pièces, 

 - Détecte la chute des pièces, 

 - Freine le moteur en fin de paiement, 

 - Protège le moteur des surcharges, 

 - Détecte le blocage d’une pièce, 

 - Libère les pièces bloquées, 

 - Sa capacité est de 1000 pièces de 2 F., 1600 pièces de 50 centimes et 

2000 pièces de 10 centimes. 

Analyse fonctionnelle : 

Liaisons fonctionnelles avec le système hôte: 

 

SEC : indicateur de fontionnement normal. 

 

MV : Magasin Vide: renvoi un état haut si le nombre de pièces est insuffisant. 

 

CA : Pièce devant le Capteur A. 

 

CB : Pièce devant le Capteur B. 

 

Demande de paiement : Demande de paiement envoyé par le système hôte au 

rendeur de monnaie. Ce signal est actif au niveau bas. Il est pris en compte si sa durée 

est supérieur à 5 ms et si l’écart entre une autre demande est supérieur à 5 ms. 

 

Liaisons fonctionnelles avec le personnel de 

maintenance: 
 

IND1 : Présence du 5 volts d’alimentation des circuits logiques. 

 

IND2 : Passage d’une pièce devant le capteur B. 

 

IND3 : Fonctionnement normal. 

 

IND4 : Passage d’une pièce devant le capteur A. 

 

Liaisons fonctionnelles internes : 
 

M1, M2, M3, M4 : commande le courant moteur entrainant la noria. 

 

IM : intensité du courant moteur. 

 

PPFS : Présence d’une Pièce devant la Fenêtre de Sortie. 

 

SI : Sur Intensité moteur. 

 

DA24 : Défaut d’Alimentation 24 volts du moteur. 



Plan de cablage 

Notice d’utilisation : 

 

Pour l’usager : 
 

 L’Usager doit poser le colis sur le système, et saisir les paramètres à l’aide 

des huit touches du clavier. Le système pèse le colis et affiche sur le moniteur le 

prix à payer, l’usager doit payer la somme voulue puis le dispositif lui rend la 

monnaie si cela est nécessaire, et une vignette d’affranchissement. 

 

 

Pour le personnel de la poste : 
 

 Le personnel de la poste doit s’assurer du bon fonctionnement du système 

grace aux indicateurs de fonctionnement normal, et vérifier que le magasin de 

pièces n’est ni vide ni plein, et qu’il reste des vignettes d’affranchissement 

vièrges dans le système. 

 En cas de panne, le technicien devra vérifier les cartes entrées sorties et 

moteur en utilisant la carte test, et contrôler les fusibles de l’alimentation et les 

tensions aux points tests. 

Deuxième Partie : 

La carte micro-contrôleur 

Sommaire : 
 
 - Présentation de la carte micro-contrôleur : 

 

  * Schéma fonctionnel de deuxième degré, 

 

  * Schéma structurel, 

 

  * Documents de fabrication : 

   - Typon 

   - Implantation des composants 

   - Nomenclature 

 

 - Documents de mise au point 

Schéma fonctionnel de 

deuxième degré : 

- Côté cuivre : 

Typon 

- Côté composants : 

Implantation des 

composants : 

Nomenclature 

Résistance : 
R1   2,2 KW 1/4W 5% couche carbone 

 

Quartz : 

Q1   11,0592 MHz 

 

Condensateurs : 
C1   22   pF 100  V Céramique 

C2   22   pF 100  V Céramique 

C3   10   µF  16   V Chimique 

C4   100 nF   63  V Plastisque 

C5   100 nF   63  V Plastisque 

C6   100 nF   63  V Plastisque 

C7   1    µF   50   V Chimique 

 

Connecteurs : 
Con1  20 HE 10 

Con2  20 HE 10 

 

Circuits intégrés : 

CI1   80C31 micro-contrôleur 

CI2   27C64 EPROM 

CI3   74 HCT 573 

 

Supports CI : 
20 Broches Double face “Tulipe” 

28 Broches Double face “Tulipe” 

40 Broches Double face “Tulipe” 

 

Divers : 
Epoxy, nappes, étain, vis, rondelles, écrous, picots 

 

         

 Total : 

          TVA : 

 

         

 TTC : 

0,09 F. H.T. 

 

 

9,00 F. H.T. 

 

 

0,66 F. H.T. 

0,66 F. H.T. 

0,66 F. H.T. 

1,50 F. H.T. 

1,50 F. H.T. 

1,50 F. H.T. 

1,00 F. H.T. 

 

 

7,08 F. H.T. 

7,08 F. H.T. 

 

 

28,75 F. H.T. 

29,00 F. H.T. 

3,32 F. H.T. 

 

 

2,10 F. H.T. 

2,20 F. H.T. 

2,40 F. H.T. 

 

 

7,50 F. H.T. 

 

106,00 F. H.T. 

21,85 F. H.T. 

 

 127,85 F. 

; ********************************************************* 

; *   programme de gestion du rendeur de monnaie *  

; ********************************************************* 

 

;Utilisation des ports 1 et 3 du 80C31 

;port1 

M1 BIT P1.0  ;coMmande T1 

M2 BIT P1.1  ;coMmande T2 

M3 BIT P1.2  ;coMmande T3 

M4 BIT P1.3  ;coMmande T4 

SI BIT P1.4  ;SurIntensité moteur 

DA24 BIT P1.5  ;Défaut Alimentation 24 V 

ALIM BIT P1.6  ;commande de l'ALIMentation 24 V 

 

;port 3 

IND1 BIT P3.0  ;INDicateur de présence alim logique 

IND2 BIT P3.1  ;INDicateur de passage pièce devant capteur 

IND3 BIT P3.2  ;INDicateur de fonctionnement normal 

DP BIT P3.3  ;Demande de Paiement; sur interruption 1 

PPFS BIT P3.4  ;Présence Pièce dans la Fenêtre de Sortie 

SEC BIT P3.5  ;SECurité 

 

;Adresses des indicateurs 

NBPIECE DATA 0H 

ETAPE DATA 1H 

SECU BIT 0H 

 

;************* Reset et interruptions ******************* 

 ORG 0H  

 LJMP  INIT  ;RESET 

 ORG 13H 

 LJMP INTER1  ;INT1 

 ORG 1BH 

 LJMP DEPASSE  ;TF1 dépassement timer 1 

 

;************* Initialisation *************************** 

 ORG 50H 

INIT: ACALL ARRET  ;arrêt moteur 

 CLR IND2 

 CLR IND3 

 CLR SEC 

 MOV NBPIECE,#0 

 MOV ETAPE,#0 

 CLR ALIM  ;validation alim 24 V 

TEST: JB DA24,TEST 

 SETB SEC 

 SETB SECU 

 SETB IND3 

 MOV TMOD,#00010001B  ;timer 1 et 0 en horloge 16 bits 

 SETB EX1  ; valide interruption 

 SETB EA  ;INT1 (active à 0) 

 

;******************* PROGRAMME PRINCIPAL ******************** 

ATTENT: JNB SECU,FREINAGE 

 MOV A,NBPIECE 

 CJNE A,#0,PIECE 

FREINAGE: ACALL FREIN 

 SJMP ATTENT 

PIECE: ACALL PAYE 

 SJMP  ATTENT 

 

;********************** Interruption 1 ********************** 

;utilisée pour l'analyse de l'information demande de paiement : DP 

 

INTER1: CLR EX1 

 CLR ET1 

 CLR TR1 

 PUSH A 

 MOV A,ETAPE 

 CJNE A,#0,ETAPE2 

 

 MOV ETAPE,#1 ; étape 1 :lancer tempo 5ms 

 MOV TL1,#0 

 MOV TH1,#0EEH ;temporisation 5 ms 

 SETB TR1 

 SETB ET1 

 SJMP FININTER1 

 

;fin de l'étape 3 : retour à l'étape 2 (niveau haut sur DP < 5ms) 

ETAPE2: MOV ETAPE,#2 

 MOV TH1,#0FFH ;pour détection niveau haut sur INT1 : 

 MOV TL1,#0F0H ;le timer est validé pour INT1 à 1 

 MOV TMOD,#10010001B ;donc dépassement quand DP à 1 

 SETB TR1 

 SETB ET1 

 

FININTER1: 

 POP A 

 RETI 

;**************** Interruption dépassement timer 1 ********************* 

;utilisée pour l'analyse de l'information demande de paiement : DP 

 

DEPASSE: CLR EX1  

 CLR ET1 

 CLR TR1 

 PUSH A 

 MOV A,ETAPE 

 CJNE A,#1,ETAPE3 

 JB DP,RETOURA0 ;fin étape 1 : si DP = 1 retour à l'étape 0, sinon étape 2 

 MOV ETAPE,#2  ;étape 2  

 MOV TH1,#0FFH ;pour détection niveau haut sur INT1 : 

 MOV TL1,#0F0H ;le timer est validé pour INT1 à 1 

 MOV TMOD,#10010001B ;donc dépassement quand DP à 1 

 SETB TR1 

 SETB ET1 

 SJMP FININTER 

RETOURA0: MOV  ETAPE,#0 

 SETB EX1 

 SJMP FININTER 

ETAPE3: MOV A,ETAPE 

 CJNE A,#2,ETAPE4 

 MOV ETAPE,#3  ;enclencher tempo 5 ms : DP à l'état haut 

 MOV TMOD,#00010001B 

 MOV TL1,#0  ; 16^4 - 5E-3 / 1,085 E-6 = 60928  

 MOV TH1,#0EEH ;    = EE00H 

 SETB TR1 

 SETB ET1 

 SETB EX1 

 SJMP FININTER 

ETAPE4: ;prendre en compte la demande de paiement et retour à l'étape 0 

 INC NBPIECE 

 MOV ETAPE,#0 

 SETB EX1 

 SJMP FININTER 

FININTER: POP A  ;fin d'interruption 

 RETI 

 

;************ Sous programme paiement d'une pièce **************** 

PAYE: ACALL MAV  ;mise en marche avant 

 ACALL TEMP100 

TESTSI: JNB SI,TESTDA ;tests jusqu'à sortie de la pièce 

 CLR IND3 

 ACALL DEBOUR 

 SETB IND3 

 SJMP TESTPPFS 

TESTDA: JNB DA24,TESTPPFS 

 ACALL FREIN 

 CLR IND3 

 CLR SEC 

 CLR SECU 

TST24: JB DA24,TST24 ;correction de l’erreur du programme 

 SETB IND3  ;fonctionnement 

 SETB SEC  ;autorisé 

 SETB SECU  

 ACALL MAV  ;après réapparition 

 ACALL TEMP100  ;du 24V 

TESTPPFS: JNB PPFS,TESTSI 

 DEC NBPIECE  ;la pièce est sortie 

 MOV A,NBPIECE 

 CJNE A,#0,SUITE 

 ACALL FREIN 

SUITE: SETB IND2 

 ACALL SCRUTE  ;test de validation de paiement 

 RET 

 

;***************** Marche avant du moteur ******************* 

MAV: ACALL ARRET 

 SETB M1 

 SETB M4 

 RET 

 

;***************** Marche arrière du moteur ******************* 

MAR: ACALL ARRET 

 SETB M2 

 SETB M3 

 ACALL TEMP100  ;temporiser 100 ms 

 MOV TH0,#4BH ;enclencher 

 MOV TL0,#0FDH ;temporisation 

 SETB TR0  ;50ms 

TEM50: JNB SI,PERSIST  

 SETB TF0  ;abréger la temporisation : surintensité 

PERSIST: JNB TF0,TEM50 

 CLR TF0 

 CLR TR0 

 JNB  SI,FINMAR 

 ACALL FREIN  ;freiner pendant 50 ms 

 ACALL ARRET  ;arrêt 

 ACALL TEMP1000 ;temporiser 1 s 

FINMAR: RET 

;***************** Freinage du moteur ******************* 

FREIN: ACALL ARRET 

 SETB M3 

 SETB M4 

 ACALL TEMP50  ;50 ms mini 

 RET 

 

;***************** Arrêt du moteur ******************* 

ARRET: CLR M1 

 CLR M2 

 CLR M3 

 CLR M4 

 ACALL TEMP1 ;1ms mini : éviter risque de conduction T1T3,T2T4 

 RET 

 

;***************** Débourrage ************************ 

DEBOUR: ACALL FREIN 

 ACALL MAR  

 ACALL FREIN 

 ACALL MAV 

 ACALL TEMP100 

 JB SI,DEBOUR 

 RET 

 

;**************** Temporisations 1 ms, 50 ms, 100 ms, 1 s *************** 

;calcul des temporisations : 

;fréquence du quartz : 11,0592 MHz d'où la période d'horloge du timer : 12/11,0592 E6 = 1,085 µs 

;le timer est chargé avec le codage hexadécimal, sur 16 bits,de la valeur décimale entière résultant du calcul : 

;16^4 - durée / 1,085 E-6 

 

;temporisation 1ms 

TEMP1: MOV TH0,#0FCH ;64614 = FC66 H 

 MOV TL0,#66H 

 SETB TR0 

BCL1: JNB TF0,BCL1 

 CLR TF0 

 CLR TR0 

 RET 

 

;temporisation 50ms 

TEMP50: MOV TH0,#4BH ;19453 = 4BFD H 

 MOV TL0,#0FDH 

 SETB TR0 

BCL50: JNB TF0,BCL50 

 CLR TF0 

 CLR TR0 

 RET 

 

;temporisation 100ms 

TEMP100: ACALL TEMP50 

 ACALL TEMP50 

 RET 

 

;temporisation 1s 

TEMP1000: MOV B,#20  ;20 fois 

BCL1000: ACALL TEMP50 

 DJNZ B,BCL1000 

 RET 

 

;*************** Scruter capteur B **************** 

;********* test de validation du paiement ********* 

SCRUTE: MOV B,#20  ;20 fois 

TEMPO50: MOV TH0,#4BH ;50 ms 

 MOV TL0,#0FDH 

 SETB TR0 

TEMP: JB PPFS,CONTINUE 

;la pièce est tombée dans la sébile 

 CLR IND2 

 SETB TF0  ;abrèger la 

 MOV B,#1  ;temporisation 

CONTINUE: JNB TF0,TEMP 

 CLR TF0 

 DJNZ B,TEMPO50 

 CLR TR0 

 JNB  IND2,FINSCRUTE 

 CLR SEC 

 CLR SECU 

 CLR IND3 

 ACALL FREIN 

DEPANNE: JB PPFS,DEPANNE ;on attend le technicien 

 CLR IND2  ;il est intervenu : 

 SETB SEC  ;la pièce est débloquée 

 SETB SECU  

 SETB IND3 

FINSCRUTE: RET 



Init ialisat ion :

DEBUT

IND2=IND3=0

SEC=NBPIECE=0

ET APE=ALIM=0

ARRET

T EMP1

DA24=1

SEC=IND3=
SECU=1

FIN

DEBUT

SECU=1

NBPIECE
=0

FREIN

ARRET

M3=M4=1

T EMP50

PAYE

Programme
principal

ALGORIGRAMME : 

